



TMS FixInsight

DEVELOPERS GUIDE

February 2020
Copyright © 2020 by tmssoftware.com bvba
Web: <http://www.tmssoftware.com>
Email: info@tmssoftware.com

Index

Index.....	2
Installation and Uninstallation.....	3
General Usage.....	4
Excluding Units from Analysis.....	6
Suppressing warnings.....	7
Command Line Tool.....	8
Code patterns handled by FixInsight.....	9
Warnings.....	9
Optimizations.....	17
Coding Conventions.....	18

Installation and Uninstallation

FixInsight can be installed by:

1. Download the installer
2. Run the installer and follow the instructions


FixInsight can be uninstalled by:

1. Open Windows Control Panel
2. Select 'Uninstall a Program'
3. Find FixInsight in the list of programs and double click on it
4. Click 'Yes' button in the FixInsight Uninstall window


General Usage

FixInsight supports RAD Studio 10.3 Rio as well as the older IDE releases (2006 – 10.2 Tokyo).


FixInsight adds menu items in the RAD Studio 'Project' menu:



In Delphi 2010 and above it also adds menu items in the project manager's context menu:




Click View->Toolbars->FixInsight to view the FixInsight toolbar:



Excluding Units from Analysis

You can exclude specific units from analysis by entering their names in the Settings window, one entry per line.



You can use file masks. Like “Jwa*.pas” or “C:\Project\ThirdParty*.pas”. Also you can ignore whole folder content: “C:\Project\FolderToIgnore”.

Another option is to add "FI:ignore" comment to the very first line of a unit that you want to skip.

Example:

```
unit Unit1; //FI:ignore
```

Suppressing warnings

You may use a special comment "FI:<RULENUMBER>" to suppress a particular warning. You need to place this comment on the line where a warning occurs.

Example:

```
procedure RestartTimer;  
begin  
    FTimer.Enabled := False;  
    FTimer.Enabled := True; //FI:W508 - this warning will be ignored.  
end;
```

The second option is to use the `_FIXINSIGHT_` compiler directive to suppress warnings. It works just like regular compiler conditional directives. Make sure that your code remains valid when `_FIXINSIGHT_` is declared and the code segment is excluded.

Example:

```
{$IFDEF _FIXINSIGHT_  
    procedure RestartTimer; // this code will be ignored  
    begin  
        FTimer.Enabled := False;  
        FTimer.Enabled := True;  
    end;  
}$ENDIF}
```

By checking "Hint on useless fi:xxx comment" option in Settings window you can get a list of lines where no warning produced by fi:xxx comment is present. This usually happens when you want to ignore a false positive but then you have installed a FixInsight update, where this false positive is fixed.

Command Line Tool

The FixInsight command line tool (FixInsightCL.exe) may be used in the build process or with continuous integration.

Parameters:

```
--project=XXX.dpr (a project to analyze)
--defines=XXX;YYY;ZZZ (compiler defines separated by semicolon)
--output=XXX (write output to a file)
--searchpath=XXX (unit directories)
--libpath=XXX (Delphi library path)
--unitaliases=XXX;YYY;ZZZ
--unitscopes=XXX;YYY;ZZZ
--ignore=XXX (units and directories to be ignored)
--settings=XXX.ficfg (override project settings)
--silent (produce no output if no issues were found)
--xml (format output as xml)
--csv (format output as csv)
```


--project parameter is mandatory, other parameters are optional.
Setting --libpath and --unitscopes parameters is highly recommended.

Example:

```
FixInsightCL -project=c:\source\example.dpr --libpath="C:\Program Files (x86)\
Embarcadero\Studio\20.0\source" --unitscopes=Winapi;System.Win;Data.Win;System
```

Some FixInsight rules will not be able to work without --libpath and --unitscopes set.

Values for --unitaliases and --unitscopes can be taken from your project options:



Code patterns handled by FixInsight

The tool produces a list of warnings when issues are found. It also checks your code for coding convention compliance.

Warnings

Potential errors and oddities.

W501 Empty EXCEPT block

The empty except block warning means the exception is caught, but not handled correctly, which in turn means the cause of the exception that occurred in the try block is still there and affecting the app is executed. Want to produce a chain reaction of errors that are impossible to trace? An empty except block is a perfect way to achieve this.

```
try
  Result := GetSomeData;
except
  // nothing
end;
```

W502 Empty FINALLY block

The finally block allows your app to free memory and finish all running process correctly. The empty finally block means the app doesn't care much about releasing handles, freeing memory and finalizing objects and therefore becomes a fertile ground for mystical elusive bugs.

```
try
  Result := GetSomeData;
finally
  // nothing
end;
```

W503 Assignment right hand side is equal to its left hand side

While such statements do not affect performance or produce errors, they are potentially dangerous if a different expression was expected on the right hand side. This type of assignment is commonly seen with copy-pasted code.

```
RowCount := RowCount;
```

W504 Missing INHERITED call in destructor

Forgetting to call an inherited destructor means the object may not destroy completely since ancestor classes' destructors are not called, resulting in memory and resource leaks at best and unexpected behaviour at worst.

```
destructor TDataContainer.Destroy;  
begin  
    FData.Free;  
end;
```

W505 Empty THEN block

One semicolon renders the entire if clause useless. The DoSomething block always executes.

```
if Param > 3 then;  
begin  
    DoSomething;  
end;
```

W506 Empty ELSE block

Similar to the above, the else block executes regardless of the value of Param. The problem with such errors is that they are extremely difficult to track down. A semicolon is easy to miss.

```
if Param > 3 then  
    DoSomething  
else;  
begin  
    Flag := -1;  
    ShowMessage('Wrong param value');  
end;
```

W507 THEN statement is equal to ELSE statement

Not a big deal if you did this intentionally (such as for debugging purposes). The check for Param effectively checks nothing and the same code is executed regardless of the check. This may indicate a logic error – that the two sides of the if/else should be different – or a copy/paste error.

```
if Param > 3 then
  DoSomething
else
  DoSomething;
```

W508 Variable is assigned twice successively

A simple error that is hard to track. Simply put, Value loses its value. Note, that the two lines of code below may be separated by 200+ lines of statements making detection of this potentially bug even harder.

```
Value := GetValue;
Value := 505;
```

W509 Unreachable code

The last line of this code block never executes. The Exit clause may reside deep in the logic of this block, so the problem becomes less evident than it is in the below example.

```
begin
  Value := GetValue;
  Exit;
  ProcessValue(Value);
end;
```

W510 Values on both sides of the operator are equal

Copy-pasting of code fragments often helps, but sometimes they can become the source of undetectable issues. Here, both operands are the same resulting in 0 being assigned to Result.

```
Result := X - X;
```

W511 Object 'Foo' created in TRY block

A common mistake by those new to the language. When an object is created inside a try block, and the application raises an exception before the object instance is assigned to the Foo variable, such as if the constructor causes an error, Foo remains unassigned and the call to Free in the finally block will try to free a random pointer. This will cause a second exception.

```
try
  Foo := TFooObject.Create;
  Foo.CallMethod;
finally
  Foo.Free;
end;
```

W512 Odd ELSE-IF condition

Why is this here? What does it do? Is the else part really needed? May be you know the answer, but more likely the values Param is checked against were different sometime in the past (or in some other code you copied this fragment from). Now they are the same and can therefore be the source of errors.

```
if Param = 1 then
  ProcA
else if Param = 1 then
  ProcB;
```

W513 Format parameter count mismatch

The number of parameters passed to Format does not match the number in the format string. This will compile but will never work.

```
Result := Format('%s = %d', [Name]);
```

W514 Loop iterator could be out of range (missing -1?)

Dynamic arrays in Delphi are indexed from 0 to Length - 1, which seems to be incorrect in this code example.

```
var
```

```
I: Integer;  
Arr: array of Char;  
begin  
  Arr := GetArr;  
  for I := 0 to Length(Arr) do  
    ProcessChar(Arr[I]);  
  end;
```

W515 Suspicious Free call

The call to Free looks suspicious in this context. For example, the code may have been intended to use 'with', but it is missing, or the entire block was copied from another part of code and then incorrectly modified. Either way, this warning indicates that the wrong object may be freed. In the below snippet, the THelloObject instance is being freed instead of the StringList instance.

```
procedure THelloObject.SaveToFile(const FileName: string);  
var  
  StringList: TStringList;  
begin  
  StringList := TStringList.Create;  
  try  
    StringList.Add('Hello world');  
    StringList.SaveToFile(FileName);  
  finally  
    Free;  
  end;  
end;
```

W517 Variable 'Foo' hides a class field, method or property

See that local Foo variable in the GetFoo method? It hides the private class member of the same name, making that class field inaccessible from within the method. In the example below, the GetFoo method will always return the value of the local Foo, not the Foo field.

```
type  
  TMyClass = class  
  private  
    Foo: Integer;  
  public
```

```
function GetFoo: Integer;  
end;  
  
function TMyClass.GetFoo: Integer;  
var  
    Foo: Integer;  
begin  
    Result := Foo;  
end;
```

W519 Method 'Foo' is empty

Empty methods are fine except they do nothing. Is this bad? Not unless you want the method to do something. Usually, empty methods are the remnants of past refactoring or changes in the class structure.

```
procedure TMyClass.Foo;  
begin  
    // nothing  
end;
```

W520 Parenthesis might be missing around IN operand

If Value is an integer it is calculated as "(not Value) in [set]", when usually what was meant is "not (Value in [set])".

```
procedure TMyClass.Foo;  
var  
    Value: Integer;  
begin  
    if not Value in [1,2,3] then  
        Bar;  
end;
```

W521 Return value of function 'Foo' might be undefined

The Delphi compiler warns you if a function may not have had its Result initialised, but only if it is a simple / unmanaged type (eg an integer). However, it doesn't do this if it is a managed type, like an interface or string.

```
function TMyClass.Foo(Param: Boolean): string;
begin
  if Param then
    Result := 'OK'
  else
    DoNotAssignResult;
end;
```

W522 Destructor without an override directive

This directive must be set, or a call to the Free method will never be successful, because the Free method calls the destructor.

W523 Interface 'Foo' declared without a GUID

If an interface does not have a GUID, it cannot be used with Supports function or with the 'as' operator.

W524 Generic interface 'Foo' declared with a GUID

This means that any all generic instantiations of this generic interface will share the same GUID. Since interfaces must have a unique GUID, and GUIDs are used for casting, this can be an major error.

```
IFoo<T> = interface
  ['{E7B8DF46-3B3D-46D3-916A-6A6008DD5B68}']
  procedure DoWork;
end;
```

W525 Missing INHERITED call in constructor

The same as rule W504, but for constructors.

```
constructor TDataContainer.Create;
begin
  FData := TList.Create;
end;
```

W526 Pointer to a nested method

When a nested method is being called from outside its "parent" method, an incorrect stack frame often leads to an error that is hard to track down.

```
procedure TMyClass.DoWork;

  procedure NestedMethod(List: TObjectList);
  var
    I: Integer;
  begin
    for I := 0 to List.Count - 1 do
      Foo(Self, List[I]);
    end;
  end;

begin
  SetCallback(@NestedMethod);
end;
```

W527 Property is referenced directly in its getter or setter

Accessing a property in its getter or setter may lead to infinite recursion.

```
type
  TTestClass = class
  private
    FProp: Integer;
    procedure SetProp(const Value: Integer);
    function GetProp: Integer;
  published
    property Prop: Integer read GetProp write SetProp;
  end;

// ...

procedure TTestClass.SetProp(const Value: Integer);
begin
  Prop := Value; // cannot assign to Prop here
end;

function TTestClass.GetProp: Integer;
begin
```

```
Result := Prop; // cannot get Prop value here  
end;
```

W528 Loop variable is not used in FOR-loop

A loop variable is not used inside the loop. This may indicate a coding error.

```
for I := 0 to 9 do  
  for J := 0 to 9 do  
    Matrix[I, I] := 0; // J is not used
```

W529 Should be 'raise' instead of 'raise object'?

When we call 'raise E;' we are telling compiler that here is a new exception object that we want to raise. After Delphi raises the exception object, the original exception object is freed.

```
procedure TForm56.Button1Click(Sender: TObject);  
  
  procedure TestProc;  
  begin  
    try  
      raise Exception.Create('FIRST EXCEPT');  
    except  
      on e: Exception do // <-- Object 'e' will be released after executing  
                          // this block  
        begin  
          ShowMessage('Except 1: ' + e.Message);  
          raise e; // <-- We should be doing this instead: 'raise;'.  
                  // Don't reference 'E' here!  
        end;  
      end;  
    end;  
  
  begin  
    try  
      TestProc;  
    except  
      on e: Exception do // <-- Object 'e' is already released in TestProc  
        begin  
          ShowMessage('Except 2: ' + e.Message);  
        end;  
      end;  
    end;
```

```
end;  
end;
```

W529 "Foo" interface has the same GUID with "Bar" (Unit1.pas)

Interface GUIDs should not be duplicated across the project.

```
IMyInterface = interface  
    [{E9FFCD4E-E7B6-4B36-B4F3-A235CA926C17}]  
end;  
  
IAnotherInterface = interface  
    [{E9FFCD4E-E7B6-4B36-B4F3-A235CA926C17}]  
end;
```

W531 Actual parameter "Foo" of FreeAndNil() must be a reference to class instance

The parameter of FreeAndNil() must be a reference to a class instance. Due to its implementation details, compiler is not able to check its parameter type.

```
var  
    I: Integer;  
    MyObject: TObject;  
begin  
    FreeAndNil(I); // This will compile, but never work  
    FreeAndNil(MyObject);  
end;
```

W534 Class instance "Foo" passed to "Bar" but interface expected

Class instance passed as interface.

```
type  
    ISomeInterface = interface  
    [{AAAAAAAA-AAAA-AAAA-AAAA-AAAAAAAAAAAA}]  
end;
```

```
TSomeClass = class(TInterfacedObject, ISomeInterface)
end;

procedure DoSomethingWithSomeInterface(const SomeInterface:
ISomeInterface);
begin
  // ...
end;

procedure DoSomethingWithSomeClass;
var
  SomeClass: TSomeClass;
begin
  // ...
  DoSomethingWithSomeInterface(SomeClass); // <= W534
  // After this call SomeClass will be released due to reference counting
  // ...
end;
```

W535 Enumerated constant(s) missing in case statement: Foo

In the code below, etThree is missing from the case structure and will trigger a warning.

```
type
  TEnumType = (etOne, etTwo, etThree);

var
  EnumType: TEnumType;
begin
  case EnumType of // <= W535
    etOne: ;
    etTwo: ;
  end;
end;
```

W536 New class instance ('Foo') passed to 'Bar' as const interface parameter.

```
type
  ISomeInterface = interface
    ['{AAAAAAAAA-AAAA-AAAA-AAAA-AAAAAAAAAAAAA}']
```

```
end;

TSomeClass = class(TInterfacedObject, ISomeInterface)
end;

procedure DoSomethingWithSomeInterface(const SomeInterface:
ISomeInterface);
begin
    // ...
end;

procedure DoSomethingWithSomeClass;
begin
    // ...
    // After this call TSomeClass instance will not be released.
    // There is a potential memory leak.
    DoSomethingWithSomeInterface(TSomeClass.Create); // <= W536
    // ...
end;
```

Optimizations

Little tricks that make your code cleaner and faster.

O801 CONST missing for unmodified string parameter 'Foo'

String parameter is not modified in a method. If you declare this parameter as a 'const' this will result in better performance. This will eliminate implicit try-finally and reference counting overhead.

```
procedure Output(S: string);  
begin  
    ShowMessage('The message is "' + S + '"');  
end;
```

O802 ResourceString 'Foo' is declared but never used

ResourceString is declared but never used. This means it may be removed without any risk.

O803 Constant 'Foo' is declared but never used

Constant is declared but never used. This means it may be removed without any risk.

O804 Method parameter "Foo" is declared but never used

```
function Output(A, B, C: Integer): Integer;  
begin // Parameter C is never used and may be removed.  
    Result := A + B;  
end;
```

O805 Inline marked routine "%s" comes after its call in the same unit

In order to be properly inlined an inline marked routine has to come before it's called in the same unit.

Coding Conventions

Coding convention violations are not technically incorrect but may slow down development or increase the risk of bugs in the future.

C101 Method 'Foo' is too long (N lines)

A method, function, or procedure that has grown too large.

C102 Too many parameters in 'Foo' (N parameters)

A long list of parameters in a procedure or function make readability and code quality worse.

C103 Too many variables in 'Foo' (N variables)

A long list of variables in a procedure or function make readability and code quality worse.

C104 Class name should start with 'T'

According to [Object Pascal Style Guide](#) class declaration should be prefaced by a capital T.

C105 Interface name should start with 'I'

According to [Object Pascal Style Guide](#) interface declaration should be prefaced by a capital I.

C106 Pointer type name should start with 'P'

According to [Object Pascal Style Guide](#) pointer type declaration should be prefaced by a capital P.

C107 Class field name should start with 'F'

According to [Object Pascal Style Guide](#) class field name should be prefaced by a capital F.

C108 Nested WITH statement

If a class's interface is changed, the behaviour of code using 'with' can change silently and without any indication. It creates imprecise semantics, and that is always bad.

```
with Obj1 do
  with Obj2 do
    DoSomething;
```

C109 Unneeded boolean comparison

Just an unnecessary comparison. Not a big deal, but a simple assignment operator looks more natural.

```
Result := A <> True;
```

C110 Getter or setter name is different from property declaration

Getter and setter method names are typically the name of the property with a 'Get' or 'Set' prefix.

```
property Caption: string read GetName write SetName;
```

C111 Class name should start with 'E'

According to [Object Pascal Style Guide](#) class declaration should be prefaced by a capital E.